# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=libmng
pkgver=2.0.3
pkgrel=3
pkgdesc="A collection of routines used to create and manipulate MNG format graphics files"
arch=('x86_64')
url="https://www.libmng.com/"
license=('custom')
depends=('lcms2')
source=(https://downloads.sourceforge.net/sourceforge/${pkgname}/${pkgname}-${pkgver}.tar.xz{,.asc})
sha512sums=('764efd94643c17c449abcb8f676ec2aa750a2461cf46bc961343f8d443a16ac2caa135c27d846deb2351b9f25d6170c42a500d21f63c13276905fdd743b8fec6'
            'SKIP')
validpgpkeys=('8048643BA2C840F4F92A195FF54984BFA16C640F')

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -D -m644 LICENSE "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}
